
Abstract
We apply Andy Pitts’s methods of defining relations over domains to several classical

results in the literature. We show that the Y combinator coincides with the domain-
theoretic fixpoint operator, that parallel-or and the Plotkin existential are not definable
in PCF, that the continuation semantics for PCF coincides with the direct semantics, and
that our domain-theoretic semantics for PCF is adequate for reasoning about contextual
equivalence in an operational semantics. Our version of PCF is untyped and has both
strict and non-strict function abstractions. The development is carried out in HOLCF.

1 Introduction

Showing the existence of relations on domains has historically been an involved process. This
is due to the presence of the contravariant function space domain constructor that defeats
familiar inductive constructions; in particular we wish to define “logical” relations, where
related functions take related arguments to related results, and the corresponding relation
transformers are not monotonic. Before Pitts (1996) such demonstrations involved laborious
appeals to the details of the domain constructions themselves. (See Mulmuley (1987); Stoy
(1977) for historical perspective.)
Here we develop some standard results about PCF using Pitts’s technique for showing the
existence of particular recursively-defined relations on domains. By doing so we demonstrate
that HOLCF (Müller et al. 1999; Huffman 2012b) is useful for reasoning about programming
language semantics and not just particular programs.
We treat a variant of the PCF language due to Plotkin (1977). It contains both call-by-name
and call-by-value abstractions and is untyped. We show the breadth of Pitts’s technique by
compiling several results, some of which have only been shown in simply-typed settings where
the existence of the logical relations is straightforward to demonstrate.

2 Pitts’s method for solving recursive domain predicates

We adopt the general theory of Pitts (1996) for solving recursive domain predicates. This is
based on the idea of minimal invariants that Pitts (1993, Def 2) ascribes “essentially to D.
Scott”.
Ideally we would like to do the proofs once and use Pitts’s relational structures. Unfortunately
it seems we need higher-order polymorphism (type functions) to make this work (but see
Huffman (2012a)). Here we develop three versions, one for each of our applications. The
proofs are similar (but not quite identical) in all cases.
We begin by defining an admissible set (aka an inclusive predicate) to be one that contains
⊥ and is closed under countable chains:
definition admS :: ′a::pcpo set set where

admS ≡ { R :: ′a set. ⊥ ∈ R ∧ adm (λx. x ∈ R) }

typedef ( ′a::pcpo) admS = { x:: ′a::pcpo set . x ∈ admS }
morphisms unlr mklr unfolding admS-def by fastforce

These sets form a complete lattice.

2.1 Sets of vectors

The simplest case involves the recursive definition of a set of vectors over a single domain.
This involves taking the fixed point of a functor where the positive (covariant) occurrences
of the recursion variable are separated from the negative (contravariant) ones. (See §3.4 etc.
for examples.)

By dually ordering the negative uses of the recursion variable the functor is made monotonic
with respect to the order on the domain ′d. Here the type constructor ′a dual yields a type
with the same elements as ′a but with the reverse order. The functions dual and undual
mediate the isomorphism.
type-synonym ′d lf-rep = ′d admS dual × ′d admS ⇒ ′d set
type-synonym ′d lf = ′d admS dual × ′d admS ⇒ ′d admS

The predicate eRSV encodes our notion of relation. (This is Pitts’s e : R ⊂ S.) We model
a vector as a function from some index type ′i to the domain ′d. Note that the minimal
invariant is for the domain ′d only.
abbreviation

eRSV :: ( ′d::pcpo → ′d) ⇒ ( ′i::type ⇒ ′d) admS dual ⇒ ( ′i ⇒ ′d) admS ⇒ bool
where

eRSV e R S ≡ ∀ d ∈ unlr (undual R). (λx. e·(d x)) ∈ unlr S

In general we can also assume that e here is strict, but we do not need to do so for our
examples.
Our locale captures the key ingredients in Pitts’s scheme:

• that the function δ is a minimal invariant;

• that the functor defining the relation is suitably monotonic; and

• that the functor is closed with respect to the minimal invariant.

locale DomSol =
fixes F :: ′a::order dual × ′a::order ⇒ ′a
assumes monoF : mono F

begin

definition sym-lr :: ′a dual × ′a ⇒ ′a dual × ′a
where

sym-lr = (λ(rm, rp). (dual (F (dual rp, undual rm)), F (rm, rp)))

lemma sym-lr-mono:
mono sym-lr

proof
fix x y :: ′a dual × ′a
obtain x1 x2 y1 y2 where [simp]: x = (x1 , x2 ) y = (y1 , y2 )

by (cases x, cases y)
assume x ≤ y
with monoF have F x ≤ F y ..
from ‹x ≤ y› have (dual y2 , undual y1 ) ≤ (dual x2 , undual x1 )

by (simp-all add: dual-less-eq-iff )
with monoF have F (dual y2 , undual y1 ) ≤ F (dual x2 , undual x1 ) ..
with ‹F x ≤ F y› show sym-lr x ≤ sym-lr y

by (simp add: sym-lr-def )
qed

end

locale DomSolV = DomSol F :: ( ′i::type ⇒ ′d::pcpo) lf for F +
fixes δ :: ( ′d::pcpo → ′d) → ′d → ′d
assumes min-inv-ID: fix·δ = ID
assumes eRSV-deltaF :∧

(e :: ′d → ′d) (R :: ( ′i ⇒ ′d) admS dual) (S :: ( ′i ⇒ ′d) admS).
eRSV e R S =⇒ eRSV (δ·e) (dual (F (dual S , undual R))) (F (R, S))

From these assumptions we can show that there is a unique object that is a solution to the
recursive equation specified by F.
definition delta ≡ delta-pos

lemma delta-sol: delta = F (dual delta, delta)
lemma delta-unique:

assumes r : F (dual r , r) = r
shows r = delta

end

We use this to show certain functions are not PCF-definable in §3.3.

2.2 Relations between domains and syntax

To show computational adequacy (§4.3) we need to relate elements of a domain to their
syntactic counterparts. An advantage of Pitts’s technique is that this is straightforward to
do.
definition synlr :: ( ′d::pcpo × ′a::type) set set where

synlr ≡ { R :: ( ′d × ′a) set. ∀ a. { d. (d, a) ∈ R } ∈ admS }

typedef ( ′d::pcpo, ′a::type) synlr = { x::( ′d × ′a) set. x ∈ synlr }
morphisms unsynlr mksynlr unfolding synlr-def by fastforce

An alternative representation (suggested by Brian Huffman) is to directly use the type ′a ⇒
′b admS as this is automatically a complete lattice. However we end up fighting the automatic
methods a lot.

Again we define functors on ( ′d, ′a) synlr.
type-synonym ( ′d, ′a) synlf-rep = ( ′d, ′a) synlr dual × ( ′d, ′a) synlr ⇒ ( ′d × ′a) set
type-synonym ( ′d, ′a) synlf = ( ′d, ′a) synlr dual × ( ′d, ′a) synlr ⇒ ( ′d, ′a) synlr

We capture our relations as before. Note we need the inclusion e to be strict for our example.
abbreviation

eRSS :: ( ′d::pcpo → ′d) ⇒ ( ′d, ′a::type) synlr dual ⇒ ( ′d, ′a) synlr ⇒ bool
where

eRSS e R S ≡ ∀ (d, a) ∈ unsynlr (undual R). (e·d, a) ∈ unsynlr S

locale DomSolSyn = DomSol F :: ( ′d::pcpo, ′a::type) synlf for F +
fixes δ :: ( ′d::pcpo → ′d) → ′d → ′d
assumes min-inv-ID: fix·δ = ID
assumes min-inv-strict:

∧
r . δ·r ·⊥ = ⊥

assumes eRS-deltaF :∧
(e :: ′d → ′d) (R :: ( ′d, ′a) synlr dual) (S :: ( ′d, ′a) synlr).
[[ e·⊥ = ⊥; eRSS e R S ]] =⇒ eRSS (δ·e) (dual (F (dual S , undual R))) (F (R, S))

Again, from these assumptions we can construct the unique solution to the recursive equation
specified by F.

2.3 Relations between pairs of domains

Following Reynolds (1974) and Filinski (2007), we want to relate two pairs of mutually-
recursive domains. Each of the pairs represents a (monadic) computation and value space.
type-synonym ( ′am, ′bm, ′av, ′bv) lr-pair = ( ′am × ′bm) admS × ( ′av × ′bv) admS

type-synonym ( ′am, ′bm, ′av, ′bv) lf-pair-rep =
( ′am, ′bm, ′av, ′bv) lr-pair dual × ( ′am, ′bm, ′av, ′bv) lr-pair ⇒ (( ′am × ′bm) set × ( ′av × ′bv)

set)

type-synonym ( ′am, ′bm, ′av, ′bv) lf-pair =
( ′am, ′bm, ′av, ′bv) lr-pair dual × ( ′am, ′bm, ′av, ′bv) lr-pair ⇒ (( ′am × ′bm) admS × ( ′av × ′bv)

admS)

The inclusions need to be strict to get our example through.
abbreviation

eRSP :: (( ′am::pcpo → ′am) × ( ′av::pcpo → ′av))
⇒ (( ′bm::pcpo → ′bm) × ( ′bv::pcpo → ′bv))
⇒ (( ′am × ′bm) admS × ( ′av × ′bv) admS) dual
⇒ ( ′am × ′bm) admS × ( ′av × ′bv) admS
⇒ bool

where
eRSP ea eb R S ≡

(∀ (am, bm) ∈ unlr (fst (undual R)). (fst ea·am, fst eb·bm) ∈ unlr (fst S))
∧ (∀ (av, bv) ∈ unlr (snd (undual R)). (snd ea·av, snd eb·bv) ∈ unlr (snd S))

locale DomSolP = DomSol F :: ( ′am::pcpo, ′bm::pcpo, ′av::pcpo, ′bv::pcpo) lf-pair for F +
fixes ad :: (( ′am → ′am) × ( ′av → ′av)) → (( ′am → ′am) × ( ′av → ′av))
fixes bd :: (( ′bm → ′bm) × ( ′bv → ′bv)) → (( ′bm → ′bm) × ( ′bv → ′bv))
assumes ad-ID: fix·ad = (ID, ID)
assumes bd-ID: fix·bd = (ID, ID)
assumes ad-strict:

∧
r . fst (ad·r)·⊥ = ⊥

∧
r . snd (ad·r)·⊥ = ⊥

assumes bd-strict:
∧

r . fst (bd·r)·⊥ = ⊥
∧

r . snd (bd·r)·⊥ = ⊥
assumes eRSP-deltaF :
[[ eRSP ea eb R S ; fst ea·⊥ = ⊥; snd ea·⊥ = ⊥; fst eb·⊥ = ⊥; snd ea·⊥ = ⊥ ]]
=⇒ eRSP (ad·ea) (bd·eb) (dual (F (dual S , undual R))) (F (R, S))

We use this solution to relate the direct and continuation semantics for PCF in §5.

3 Logical relations for definability in PCF

Using this machinery we can demonstrate some classical results about PCF (Plotkin 1977).
We diverge from the traditional treatment by considering PCF as an untyped language and
including both call-by-name (CBN) and call-by-value (CBV) abstractions following Reynolds
(1974). We also adopt some of the presentation of Winskel (1993, Chapter 11), in particular
by making the fixed point operator a binding construct.

We model the syntax of PCF as a HOL datatype, where variables have names drawn from
the naturals:
type-synonym var = nat

datatype expr =
Var var

| App expr expr
| AbsN var expr
| AbsV var expr
| Diverge (‹Ω›)
| Fix var expr
| tt
| ff
| Cond expr expr expr
| Num nat
| Succ expr
| Pred expr
| IsZero expr

3.1 Direct denotational semantics

We give this language a direct denotational semantics by interpreting it into a domain of
values.
domain ValD =

ValF (lazy appF :: ValD → ValD)
| ValTT | ValFF
| ValN (lazy nat)

The lazy keyword means that the ValF constructor is lifted, i.e. ValF ·⊥ 6= ⊥, which further
means that ValF ·(Λ x. ⊥) 6= ⊥.
The naturals are discretely ordered.

The minimal invariant for ValD is straightforward; the function cfun-map·f ·g·h denotes g oo
h oo f.
fixrec

ValD-copy-rec :: (ValD → ValD) → (ValD → ValD)
where

ValD-copy-rec·r ·(ValF ·f ) = ValF ·(cfun-map·r ·r ·f )
| ValD-copy-rec·r ·(ValTT ) = ValTT
| ValD-copy-rec·r ·(ValFF) = ValFF
| ValD-copy-rec·r ·(ValN ·n) = ValN ·n

We interpret the PCF constants in the obvious ways. “Ill-typed” uses of these combinators
are mapped to ⊥.
definition cond :: ValD → ValD → ValD → ValD where

cond ≡ Λ i t e. case i of ValF ·f ⇒ ⊥ | ValTT ⇒ t | ValFF ⇒ e | ValN ·n ⇒ ⊥

definition succ :: ValD → ValD where
succ ≡ Λ (ValN ·n). ValN ·(n + 1 )

definition pred :: ValD → ValD where
pred ≡ Λ (ValN ·n). case n of 0 ⇒ ⊥ | Suc n ⇒ ValN ·n

definition isZero :: ValD → ValD where
isZero ≡ Λ (ValN ·n). if n = 0 then ValTT else ValFF

We model environments simply as continuous functions from variable names to values.
type-synonym Var = var
type-synonym ′a Env = Var → ′a

definition env-empty :: ′a Env where
env-empty ≡ ⊥

definition env-ext :: Var → ′a → ′a Env → ′a Env where
env-ext ≡ Λ v x % v ′. if v = v ′ then x else %·v ′

The semantics is given by a function defined by primitive recursion over the syntax.
type-synonym EnvD = ValD Env

primrec
evalD :: expr ⇒ EnvD → ValD

where
evalD (Var v) = (Λ %. %·v)

| evalD (App f x) = (Λ %. appF ·(evalD f ·%)·(evalD x·%))
| evalD (AbsN v e) = (Λ %. ValF ·(Λ x. evalD e·(env-ext·v·x·%)))
| evalD (AbsV v e) = (Λ %. ValF ·(strictify·(Λ x. evalD e·(env-ext·v·x·%))))
| evalD (Diverge) = (Λ %. ⊥)
| evalD (Fix v e) = (Λ %. µ x. evalD e·(env-ext·v·x·%))
| evalD (tt) = (Λ %. ValTT )
| evalD (ff ) = (Λ %. ValFF)
| evalD (Cond i t e) = (Λ %. cond·(evalD i·%)·(evalD t·%)·(evalD e·%))
| evalD (Num n) = (Λ %. ValN ·n)
| evalD (Succ e) = (Λ %. succ·(evalD e·%))
| evalD (Pred e) = (Λ %. pred·(evalD e·%))
| evalD (IsZero e) = (Λ %. isZero·(evalD e·%))

abbreviation eval ′ :: expr ⇒ ValD Env ⇒ ValD (‹[[-]]-› [0 ,1000 ] 60 ) where
eval ′ M % ≡ evalD M ·%

3.2 The Y Combinator

We can shown the Y combinator is the least fixed point operator using just the minimal
invariant. In other words, fix is definable in untyped PCF minus the Fix construct.
This is Example 3.6 from Pitts (1996). He attributes the proof to Plotkin.
These two functions are ∆ ≡ λf x. f (x x) and Y ≡ λf . (∆ f ) (∆ f ).
Note the numbers here are names, not de Bruijn indices.
definition Y-delta :: expr where

Y-delta ≡ AbsN 0 (AbsN 1 (App (Var 0 ) (App (Var 1 ) (Var 1 ))))

definition Ycomb :: expr where
Ycomb ≡ AbsN 0 (App (App Y-delta (Var 0 )) (App Y-delta (Var 0 )))

definition fixD :: ValD → ValD where
fixD ≡ Λ (ValF ·f ). fix·f

lemma Y : [[Ycomb]]% = ValF ·fixD

3.3 Logical relations for definability

An element of ValD is definable if there is an expression that denotes it.
definition definable :: ValD ⇒ bool where

definable d ≡ ∃M . [[M ]]env-empty = d

A classical result about PCF is that while the denotational semantics is adequate, as we show
in §4, it is not fully abstract, i.e. it contains undefinable values (junk).
One way of showing this is to reason operationally; see, for instance, Plotkin (1977, §4) and
Gunter (1992, §6.1).
Another is to use logical relations, following Plotkin (1973), and also Mitchell (1996); Sieber
(1992); Stoughton (1993).
For this purpose we define a logical relation to be a set of vectors over ValD that is closed
under continuous functions of type ValD → ValD. This is complicated by the ValF tag and
having strict function abstraction.
definition

logical-relation :: ( ′i::type ⇒ ValD) set ⇒ bool
where

logical-relation R ≡
(∀ fs ∈ R. ∀ xs ∈ R. (λj. appF ·(fs j)·(xs j)) ∈ R)

∧ (∀ fs ∈ R. ∀ xs ∈ R. (λj. strictify·(appF ·(fs j))·(xs j)) ∈ R)
∧ (∀ fs. (∀ xs ∈ R. (λj. (fs j)·(xs j)) ∈ R) −→ (λj. ValF ·(fs j)) ∈ R)
∧ (∀ fs. (∀ xs ∈ R. (λj. strictify·(fs j)·(xs j)) ∈ R) −→ (λj. ValF ·(strictify·(fs j))) ∈ R)
∧ (∀ xs ∈ R. (λj. fixD·(xs j)) ∈ R)
∧ (∀ cs ∈ R. ∀ ts ∈ R. ∀ es ∈ R. (λj. cond·(cs j)·(ts j)·(es j)) ∈ R)
∧ (∀ xs ∈ R. (λj. succ·(xs j)) ∈ R)
∧ (∀ xs ∈ R. (λj. pred·(xs j)) ∈ R)
∧ (∀ xs ∈ R. (λj. isZero·(xs j)) ∈ R)

In the context of PCF these relations also need to respect the constants.
definition

PCF-consts-rel :: ( ′i::type ⇒ ValD) set ⇒ bool
where

PCF-consts-rel R ≡
⊥ ∈ R

∧ (λi. ValTT ) ∈ R
∧ (λi. ValFF) ∈ R
∧ (∀n. (λi. ValN ·n) ∈ R)

abbreviation
PCF-lr R ≡ adm (λx. x ∈ R) ∧ logical-relation R ∧ PCF-consts-rel R

The fundamental property of logical relations states that all PCF expressions satisfy all PCF
logical relations. This result is essentially due to Plotkin (1973). The proof is by a straight-
forward induction on the expression M.

lemma lr-fundamental:
assumes lr : PCF-lr R
assumes %: ∀ v. (λi. % i·v) ∈ R
shows (λi. [[M ]](% i)) ∈ R

We can use this result to show that there is no PCF term that maps the vector args ∈ R to
result /∈ R for some logical relation R. If we further show that there is a function f in ValD
such that f args = result then we can conclude that f is not definable.
abbreviation

appFLv :: ValD ⇒ ( ′i::type ⇒ ValD) list ⇒ ( ′i ⇒ ValD)
where

appFLv f args ≡ (λi. foldl (λf x. appF ·f ·(x i)) f args)

lemma lr-appFLv:
assumes lr : logical-relation R
assumes f : (λi:: ′i::type. f ) ∈ R
assumes args: set args ⊆ R
shows appFLv f args ∈ R

corollary not-definable:
fixes R :: ( ′i::type ⇒ ValD) set
fixes args :: ( ′i ⇒ ValD) list
fixes result :: ′i ⇒ ValD
assumes lr : PCF-lr R
assumes args: set args ⊆ R
assumes result: result /∈ R
shows ¬(∃ (f ::ValD). definable f ∧ appFLv f args = result)

3.4 Parallel OR is not definable

We show that parallel-or is not λ-definable following Sieber (1992) and Stoughton (1993).
Parallel-or is similar to the familiar short-circuting or except that if the first argument is ⊥
and the second one is ValTT, we get ValTT (and not ⊥). It is continuous and then have
included in the ValD domain.
definition por :: ValD ⇒ ValD ⇒ ValD (‹- por -› [31 ,30 ] 30 ) where

x por y ≡
if x = ValTT then ValTT

else if y = ValTT then ValTT
else if (x = ValFF ∧ y = ValFF) then ValFF else ⊥

The defining properties of parallel-or.
lemma POR-simps [simp]:
(ValTT por y) = ValTT
(x por ValTT ) = ValTT
(ValFF por ValFF) = ValFF
(ValFF por ⊥) = ⊥
(ValFF por ValN ·n) = ⊥
(ValFF por ValF ·f ) = ⊥
(⊥ por ValFF) = ⊥
(ValN ·n por ValFF) = ⊥
(ValF ·f por ValFF) = ⊥

(⊥ por ⊥) = ⊥
(⊥ por ValN ·n) = ⊥
(⊥ por ValF ·f ) = ⊥
(ValN ·n por ⊥) = ⊥
(ValF ·f por ⊥) = ⊥
(ValN ·m por ValN ·n) = ⊥
(ValN ·n por ValF ·f ) = ⊥
(ValF ·f por ValN ·n) = ⊥
(ValF ·f por ValF ·g) = ⊥
unfolding por-def by simp-all

We need three-element vectors.
datatype Three = One | Two | Three

The standard logical relation R that demonstrates POR is not definable is:

(x, y, z) ∈ R iff x = y = z ∨ (x = ⊥ ∨ y = ⊥)

That POR satisfies this relation can be seen from its truth table (see below).
Note we restrict the x = y = z clause to non-function values. Adding functions breaks the
“logical relations” property.
definition

POR-base-lf-rep :: (Three ⇒ ValD) lf-rep
where

POR-base-lf-rep ≡ λ(mR, pR).
{ (λi. ValTT ) } ∪ { (λi. ValFF) } — x = y = z for bools

∪ (
⋃

n. { (λi. ValN ·n) }) — x = y = z for numerals
∪ { f . f One = ⊥ } — x = ⊥
∪ { f . f Two = ⊥ } — y = ⊥

We close this relation with respect to continuous functions. This functor yields an admissible
relation for all r and is monotonic.
definition

fn-lf-rep :: ( ′i::type ⇒ ValD) lf-rep
where

fn-lf-rep ≡ λ(mR, pR). { λi. ValF ·(fs i) |fs. ∀ xs ∈ unlr (undual mR). (λj. (fs j)·(xs j)) ∈ unlr pR }

definition POR-lf-rep :: (Three ⇒ ValD) lf-rep where
POR-lf-rep R ≡ POR-base-lf-rep R ∪ fn-lf-rep R

abbreviation POR-lf ≡ λr . mklr (POR-lf-rep r)

Again it yields an admissible relation and is monotonic.
We need to show the functor respects the minimal invariant.
lemma min-inv-POR-lf :

assumes eRSV e R ′ S ′

shows eRSV (ValD-copy-rec·e) (dual (POR-lf (dual S ′, undual R ′))) (POR-lf (R ′, S ′))

We can show that the solution satisfies the expectations of the fundamental theorem lr-fundamental.
lemma PCF-lr-POR-delta: PCF-lr (unlr POR.delta)

This is the truth-table for POR rendered as a vector: we seek a function that simultaneously
maps the two argument vectors to the result.
definition POR-arg1-rel where

POR-arg1-rel ≡ λi. case i of One ⇒ ValTT | Two ⇒ ⊥ | Three ⇒ ValFF

definition POR-arg2-rel where
POR-arg2-rel ≡ λi. case i of One ⇒ ⊥ | Two ⇒ ValTT | Three ⇒ ValFF

definition POR-result-rel where
POR-result-rel ≡ λi. case i of One ⇒ ValTT | Two ⇒ ValTT | Three ⇒ ValFF

lemma lr-POR-arg1-rel: POR-arg1-rel ∈ unlr POR.delta
unfolding POR-arg1-rel-def by auto

lemma lr-POR-arg2-rel: POR-arg2-rel ∈ unlr POR.delta
unfolding POR-arg2-rel-def by auto

lemma lr-POR-result-rel: POR-result-rel /∈ unlr POR.delta

Parallel-or satisfies these tests:
theorem POR-sat:

appFLv (ValF ·(Λ x. ValF ·(Λ y. x por y))) [POR-arg1-rel, POR-arg2-rel] = POR-result-rel
unfolding POR-arg1-rel-def POR-arg2-rel-def POR-result-rel-def
by (simp add: fun-eq-iff split: Three.splits)

... but is not PCF-definable:
theorem POR-is-not-definable:

shows ¬(∃ f . definable f ∧ appFLv f [POR-arg1-rel, POR-arg2-rel] = POR-result-rel)
apply (rule not-definable[where R=unlr POR.delta])

using lr-POR-arg1-rel lr-POR-arg2-rel lr-POR-result-rel PCF-lr-POR-delta
apply simp-all

done

3.5 Plotkin’s existential quantifier

We can also show that the existential quantifier of Plotkin (1977, §5) is not PCF-definable
using logical relations.
Our definition is quite loose; if the argument function f maps any value to ValTT then
plotkin-exists yields ValTT. It may be more plausible to test f on numerals only.
definition plotkin-exists :: ValD ⇒ ValD where

plotkin-exists f ≡
if (appF ·f ·⊥ = ValFF)

then ValFF
else if (∃n. appF ·f ·n = ValTT ) then ValTT else ⊥

We can show this function is continuous.
lemma cont-pe [cont2cont, simp]: cont plotkin-exists

Again we construct argument and result test vectors such that plotkin-exists satisfies these
tests but no PCF-definable term does.

definition PE-arg-rel where
PE-arg-rel ≡ λi. ValF ·(case i of

0 ⇒ (Λ -. ValFF)
| Suc n ⇒ (Λ (ValN ·x). if x = Suc n then ValTT else ⊥))

definition PE-result-rel where
PE-result-rel ≡ λi. case i of 0 ⇒ ValFF | Suc n ⇒ ValTT

Note that unlike the POR case the argument relation does not characterise PE: we don’t treat
functions that return ValTTs and ValFFs.
The Plotkin existential satisfies these tests:
theorem pe-sat:

appFLv (ValF ·(Λ x. plotkin-exists x)) [PE-arg-rel] = PE-result-rel
unfolding PE-arg-rel-def PE-result-rel-def
by (clarsimp simp: fun-eq-iff split: nat.splits)

As for POR, the difference between the two vectors is that the argument can diverge but not
the result.
definition PE-base-lf-rep :: (nat ⇒ ValD) lf-rep where

PE-base-lf-rep ≡ λ(mR, pR).
{ ⊥ }

∪ { (λi. ValTT ) } ∪ { (λi. ValFF) } — x = y = z for bools
∪ (

⋃
n. { (λi. ValN ·n) }) — x = y = z for numerals

∪ { f . f 1 = ⊥ ∨ f 2 = ⊥ } — Vectors that diverge on one or two.

Again we close this under the function space, and show that it is admissible, monotonic and
respects the minimal invariant.
definition PE-lf-rep :: (nat ⇒ ValD) lf-rep where

PE-lf-rep R ≡ PE-base-lf-rep R ∪ fn-lf-rep R

abbreviation PE-lf ≡ λr . mklr (PE-lf-rep r)

The solution satisfies the expectations of the fundamental theorem:
lemma PCF-lr-PE-delta: PCF-lr (unlr PE .delta)
lemma lr-PE-arg-rel: PE-arg-rel ∈ unlr PE .delta
lemma lr-PE-result-rel: PE-result-rel /∈ unlr PE .delta
theorem PE-is-not-definable: ¬(∃ f . definable f ∧ appFLv f [PE-arg-rel] = PE-result-rel)

3.6 Concluding remarks

These techniques could be used to show that Haskell’s seq operation is not PCF-definable.
(It is definable for each base “type” separately, and requires some care on function values.)
If we added an (unlifted) product type then it should be provable that parallel evaluation is
required to support seq on these objects (given seq on all other objects). (See Hudak et al.
(2007, §5.4) and sundry posts to the internet by Lennart Augustsson.) This may be difficult
to do plausibly without adding a type system.

4 Logical relations for computational adequacy

We relate the denotational semantics for PCF of §3.1 to a big-step (or natural) operational
semantics. This follows Pitts (1993).

4.1 Direct semantics using de Bruijn notation

In contrast to §3 we must be more careful in our treatment of α-equivalent terms, as we
would like our operational semantics to identify of all these. To that end we adopt de Bruijn
notation, adapting the work of Nipkow (2001), and show that it is suitably equivalent to our
original syntactic story.
datatype db =

DBVar var
| DBApp db db
| DBAbsN db
| DBAbsV db
| DBDiverge
| DBFix db
| DBtt
| DBff
| DBCond db db db
| DBNum nat
| DBSucc db
| DBPred db
| DBIsZero db

Nipkow et al’s substitution operation is defined for arbitrary open terms. In our case we only
substitute closed terms into terms where only the variable 0 may be free, and while we could
develop a simpler account, we retain the traditional one.
fun

lift :: db ⇒ nat ⇒ db
where

lift (DBVar i) k = DBVar (if i < k then i else (i + 1 ))
| lift (DBAbsN s) k = DBAbsN (lift s (k + 1 ))
| lift (DBAbsV s) k = DBAbsV (lift s (k + 1 ))
| lift (DBApp s t) k = DBApp (lift s k) (lift t k)
| lift (DBFix e) k = DBFix (lift e (k + 1 ))
| lift (DBCond c t e) k = DBCond (lift c k) (lift t k) (lift e k)
| lift (DBSucc e) k = DBSucc (lift e k)
| lift (DBPred e) k = DBPred (lift e k)
| lift (DBIsZero e) k = DBIsZero (lift e k)
| lift x k = x

fun
subst :: db ⇒ db ⇒ var ⇒ db (‹-<- ′/->› [300 , 0 , 0 ] 300 )

where
subst-Var : (DBVar i)<s/k> =

(if k < i then DBVar (i – 1 ) else if i = k then s else DBVar i)
| subst-AbsN : (DBAbsN t)<s/k> = DBAbsN (t<lift s 0 / k+1>)
| subst-AbsV : (DBAbsV t)<s/k> = DBAbsV (t<lift s 0 / k+1>)
| subst-App: (DBApp t u)<s/k> = DBApp (t<s/k>) (u<s/k>)

| (DBFix e)<s/k> = DBFix (e<lift s 0 / k+1>)
| (DBCond c t e)<s/k> = DBCond (c<s/k>) (t<s/k>) (e<s/k>)
| (DBSucc e)<s/k> = DBSucc (e<s/k>)
| (DBPred e)<s/k> = DBPred (e<s/k>)
| (DBIsZero e)<s/k> = DBIsZero (e<s/k>)
| subst-Consts: x<s/k> = x

We elide the standard lemmas about these operations.
A variable is free in a de Bruijn term in the standard way.
fun

freedb :: db ⇒ var ⇒ bool
where

freedb (DBVar j) k = (j = k)
| freedb (DBAbsN s) k = freedb s (k + 1 )
| freedb (DBAbsV s) k = freedb s (k + 1 )
| freedb (DBApp s t) k = (freedb s k ∨ freedb t k)
| freedb (DBFix e) k = freedb e (Suc k)
| freedb (DBCond c t e) k = (freedb c k ∨ freedb t k ∨ freedb e k)
| freedb (DBSucc e) k = freedb e k
| freedb (DBPred e) k = freedb e k
| freedb (DBIsZero e) k = freedb e k
| freedb - - = False

Programs are closed expressions.
definition closed :: db ⇒ bool where

closed e ≡ ∀ i. ¬ freedb e i

The direct denotational semantics is almost identical to that given in §3.1, apart from this
change in the representation of environments.
definition env-empty-db :: ′a Env where

env-empty-db ≡ ⊥

definition env-ext-db :: ′a → ′a Env → ′a Env where
env-ext-db ≡ Λ x % v. (case v of 0 ⇒ x | Suc v ′ ⇒ %·v ′)

primrec
evalDdb :: db ⇒ ValD Env → ValD

where
evalDdb (DBVar i) = (Λ %. %·i)

| evalDdb (DBApp f x) = (Λ %. appF ·(evalDdb f ·%)·(evalDdb x·%))
| evalDdb (DBAbsN e) = (Λ %. ValF ·(Λ x. evalDdb e·(env-ext-db·x·%)))
| evalDdb (DBAbsV e) = (Λ %. ValF ·(strictify·(Λ x. evalDdb e·(env-ext-db·x·%))))
| evalDdb (DBDiverge) = (Λ %. ⊥)
| evalDdb (DBFix e) = (Λ %. µ x. evalDdb e·(env-ext-db·x·%))
| evalDdb (DBtt) = (Λ %. ValTT )
| evalDdb (DBff ) = (Λ %. ValFF)
| evalDdb (DBCond c t e) = (Λ %. cond·(evalDdb c·%)·(evalDdb t·%)·(evalDdb e·%))
| evalDdb (DBNum n) = (Λ %. ValN ·n)
| evalDdb (DBSucc e) = (Λ %. succ·(evalDdb e·%))
| evalDdb (DBPred e) = (Λ %. pred·(evalDdb e·%))
| evalDdb (DBIsZero e) = (Λ %. isZero·(evalDdb e·%))

We show that our direct semantics using de Bruijn notation coincides with the evaluator of §3
by translating between the syntaxes and showing that the evaluators yield identical results.
Firstly we show how to translate an expression using names into a nameless term. The
following function finds the first mention of a variable in a list of variables.
primrec index :: var list ⇒ var ⇒ nat ⇒ nat where

index [] v n = n
| index (h # t) v n = (if v = h then n else index t v (Suc n))

primrec
transdb :: expr ⇒ var list ⇒ db

where
transdb (Var i) Γ = DBVar (index Γ i 0 )

| transdb (App t1 t2 ) Γ = DBApp (transdb t1 Γ) (transdb t2 Γ)
| transdb (AbsN v t) Γ = DBAbsN (transdb t (v # Γ))
| transdb (AbsV v t) Γ = DBAbsV (transdb t (v # Γ))
| transdb (Diverge) Γ = DBDiverge
| transdb (Fix v e) Γ = DBFix (transdb e (v # Γ))
| transdb (tt) Γ = DBtt
| transdb (ff ) Γ = DBff
| transdb (Cond c t e) Γ = DBCond (transdb c Γ) (transdb t Γ) (transdb e Γ)
| transdb (Num n) Γ = (DBNum n)
| transdb (Succ e) Γ = DBSucc (transdb e Γ)
| transdb (Pred e) Γ = DBPred (transdb e Γ)
| transdb (IsZero e) Γ = DBIsZero (transdb e Γ)

This semantics corresponds with the direct semantics for named expressions.
lemma evalD-evalDdb:

assumes free e = []
shows [[e]]% = evalDdb (transdb e [])·%
using assms by (simp add: evalD-evalDdb-open)

Conversely, all de Bruijn expressions have named equivalents.
primrec

transdb-inv :: db ⇒ (var ⇒ var) ⇒ var ⇒ var ⇒ expr
where

transdb-inv (DBVar i) Γ c k = Var (Γ i)
| transdb-inv (DBApp t1 t2 ) Γ c k = App (transdb-inv t1 Γ c k) (transdb-inv t2 Γ c k)
| transdb-inv (DBAbsN e) Γ c k = AbsN (c + k) (transdb-inv e (case-nat (c + k) Γ) c (k + 1 ))
| transdb-inv (DBAbsV e) Γ c k = AbsV (c + k) (transdb-inv e (case-nat (c + k) Γ) c (k + 1 ))
| transdb-inv (DBDiverge) Γ c k = Diverge
| transdb-inv (DBFix e) Γ c k = Fix (c + k) (transdb-inv e (case-nat (c + k) Γ) c (k + 1 ))
| transdb-inv (DBtt) Γ c k = tt
| transdb-inv (DBff ) Γ c k = ff
| transdb-inv (DBCond i t e) Γ c k =

Cond (transdb-inv i Γ c k) (transdb-inv t Γ c k) (transdb-inv e Γ c k)
| transdb-inv (DBNum n) Γ c k = (Num n)
| transdb-inv (DBSucc e) Γ c k = Succ (transdb-inv e Γ c k)
| transdb-inv (DBPred e) Γ c k = Pred (transdb-inv e Γ c k)
| transdb-inv (DBIsZero e) Γ c k = IsZero (transdb-inv e Γ c k)

lemma transdb-inv:

assumes closed e
shows transdb (transdb-inv e Γ c k) Γ ′ = e

4.2 Operational Semantics

The evaluation relation (big-step, or natural operational semantics). This is similar to Gunter
(1992, §6.2), Pitts (1993) and Winskel (1993, Chapter 11).
We firstly define the values that expressions can evaluate to: these are either constants or
closed abstractions.
inductive

val :: db ⇒ bool
where

v-Num[intro]: val (DBNum n)
| v-FF [intro]: val DBff
| v-TT [intro]: val DBtt
| v-AbsN [intro]: val (DBAbsN e)
| v-AbsV [intro]: val (DBAbsV e)

inductive
evalOP :: db ⇒ db ⇒ bool (‹- ⇓ -› [50 ,50 ] 50 )

where
evalOP-AppN [intro]: [[ P ⇓ DBAbsN M ; M<Q/0> ⇓ V ]] =⇒ DBApp P Q ⇓ V

| evalOP-AppV [intro]: [[ P ⇓ DBAbsV M ; Q ⇓ q; M<q/0> ⇓ V ]] =⇒ DBApp P Q ⇓ V
| evalOP-AbsN [intro]: val (DBAbsN e) =⇒ DBAbsN e ⇓ DBAbsN e
| evalOP-AbsV [intro]: val (DBAbsV e) =⇒ DBAbsV e ⇓ DBAbsV e
| evalOP-Fix[intro]: P<DBFix P/0> ⇓ V =⇒ DBFix P ⇓ V
| evalOP-tt[intro]: DBtt ⇓ DBtt
| evalOP-ff [intro]: DBff ⇓ DBff
| evalOP-CondTT [intro]: [[ C ⇓ DBtt; T ⇓ V ]] =⇒ DBCond C T E ⇓ V
| evalOP-CondFF [intro]: [[ C ⇓ DBff ; E ⇓ V ]] =⇒ DBCond C T E ⇓ V
| evalOP-Num[intro]: DBNum n ⇓ DBNum n
| evalOP-Succ[intro]: P ⇓ DBNum n =⇒ DBSucc P ⇓ DBNum (Suc n)
| evalOP-Pred[intro]: P ⇓ DBNum (Suc n) =⇒ DBPred P ⇓ DBNum n
| evalOP-IsZeroTT [intro]: [[ E ⇓ DBNum 0 ]] =⇒ DBIsZero E ⇓ DBtt
| evalOP-IsZeroFF [intro]: [[ E ⇓ DBNum n; 0 < n ]] =⇒ DBIsZero E ⇓ DBff

It is straightforward to show that this relation is deterministic and sound with respect to the
denotational semantics.
theorem evalOP-sound:

assumes P ⇓ V
shows evalDdb P·% = evalDdb V ·%

We can use soundness to conclude that POR is not definable operationally either. We rely
on transdb-inv to map our de Bruijn term into the syntactic universe of §3 and appeal to the
results of §3.4. This takes some effort as ValD contains irrelevant junk that makes it hard to
draw obvious conclusions; we use DBCond to restrict the arguments to the putative witness.
definition

isPORdb e ≡ closed e
∧ DBApp (DBApp e DBtt) DBDiverge ⇓ DBtt
∧ DBApp (DBApp e DBDiverge) DBtt ⇓ DBtt
∧ DBApp (DBApp e DBff ) DBff ⇓ DBff

lemma POR-is-not-operationally-definable: ¬isPORdb e

4.3 Computational Adequacy

The lemma evalOP-sound tells us that the operational semantics preserves the denotational
semantics. We might also hope that the two are somehow equivalent, but due to the junk in
the domain-theoretic model (see §3.3) we cannot expect this to be entirely straightforward.
Here we show that the denotational semantics is computationally adequate, which means that
it can be used to soundly reason about contextual equivalence.
We follow Pitts (1993, 1996) by defining a suitable logical relation between our ValD domain
and the set of programs (closed terms). These are termed "formal approximation relations"
by Plotkin. The machinery of §2.2 requires us to define a unique bottom element, which in
this case is {⊥} × {P. closed P}. To that end we define the type of programs.
typedef Prog = { P. closed P }

morphisms unProg mkProg by fastforce

definition
ca-lf-rep :: (ValD, Prog) synlf-rep

where
ca-lf-rep ≡ λ(rm, rp).

({⊥} × UNIV )
∪ { (d, P) |d P.

(∃n. d = ValN ·n ∧ unProg P ⇓ DBNum n)
∨ (d = ValTT ∧ unProg P ⇓ DBtt)
∨ (d = ValFF ∧ unProg P ⇓ DBff )
∨ (∃ f M . d = ValF ·f ∧ unProg P ⇓ DBAbsN M

∧ (∀ (x, X) ∈ unsynlr (undual rm). (f ·x, mkProg (M<unProg X/0>)) ∈ unsynlr rp))
∨ (∃ f M . d = ValF ·f ∧ unProg P ⇓ DBAbsV M ∧ f ·⊥ = ⊥

∧ (∀ (x, X) ∈ unsynlr (undual rm). ∀V . unProg X ⇓ V
−→ (f ·x, mkProg (M<V /0>)) ∈ unsynlr rp)) }

abbreviation ca-lr :: (ValD, Prog) synlf where
ca-lr ≡ λr . mksynlr (ca-lf-rep r)

Intuitively we relate domain-theoretic values to all programs that converge to the correspond-
ing syntatic values. If a program has a non-⊥ denotation then we can use this relation to
conclude something about the value it (operationally) converges to.

interpretation ca: DomSolSyn ca-lr ValD-copy-rec
apply standard

apply (rule mono-ca-lr)
apply (rule ValD-copy-ID)

apply simp
apply (erule (1 ) min-inv-ca-lr)
done

definition ca-lr-syn :: ValD ⇒ db ⇒ bool (‹- / -› [80 ,80 ] 80 ) where
d / P ≡ (d, P) ∈ { (x, unProg Y ) |x Y . (x, Y ) ∈ unsynlr ca.delta }

To establish this result we need a “closing substitution” operation. It seems easier to define
it directly in this simple-minded way than reusing the standard substitution operation.

This is quite similar to a context-plugging (non-capturing) substitution operation, where the
“holes” are free variables, and indeed we use it as such below.
fun

closing-subst :: db ⇒ (var ⇒ db) ⇒ var ⇒ db
where

closing-subst (DBVar i) Γ k = (if k ≤ i then Γ (i – k) else DBVar i)
| closing-subst (DBApp t u) Γ k = DBApp (closing-subst t Γ k) (closing-subst u Γ k)
| closing-subst (DBAbsN t) Γ k = DBAbsN (closing-subst t Γ (k + 1 ))
| closing-subst (DBAbsV t) Γ k = DBAbsV (closing-subst t Γ (k + 1 ))
| closing-subst (DBFix e) Γ k = DBFix (closing-subst e Γ (k + 1 ))
| closing-subst (DBCond c t e) Γ k =

DBCond (closing-subst c Γ k) (closing-subst t Γ k) (closing-subst e Γ k)
| closing-subst (DBSucc e) Γ k = DBSucc (closing-subst e Γ k)
| closing-subst (DBPred e) Γ k = DBPred (closing-subst e Γ k)
| closing-subst (DBIsZero e) Γ k = DBIsZero (closing-subst e Γ k)
| closing-subst x Γ k = x

We can show it has the expected properties when all terms in Γ are closed.

The key lemma is shown by induction over e for arbitrary environments (Γ and %):
lemma ca-open:

assumes ∀ v. freedb e v −→ %·v / Γ v ∧ closed (Γ v)
shows evalDdb e·% / closing-subst e Γ 0

lemma ca-closed:
assumes closed e
shows evalDdb e·env-empty-db / e
using ca-open[where e=e and %=env-empty-db] assms
by (simp add: closed-def )

theorem ca:
assumes nb: evalDdb e·env-empty-db 6= ⊥
assumes closed e
shows ∃V . e ⇓ V
using ca-closed[OF ‹closed e›] nb
by (auto elim!: ca-lrE)

This last result justifies reasoning about contextual equivalence using the denotational se-
mantics, as we now show.

4.3.1 Contextual Equivalence

As we are using an un(i)typed language, we take a context C to be an arbitrary term, where
the free variables are the “holes”. We substitute a closed expression e uniformly for all of the
free variables in C. If open, the term e can be closed using enough AbsN s. This seems to be a
standard trick now, see e.g. Koutavas and Wand (2006). If we didn’t have CBN (only CBV)
then it might be worth showing that this is an adequate treatment.
definition ctxt-sub :: db ⇒ db ⇒ db (‹(-<->)› [300 , 0 ] 300 ) where

C<e> ≡ closing-subst C (λ-. e) 0

Following Pitts (1996) we define a relation between values that “have the same form”. This
is weak at functional values. We don’t distinguish between strict and non-strict abstractions.

inductive
have-the-same-form :: db ⇒ db ⇒ bool (‹- ∼ -› [50 ,50 ] 50 )

where
DBAbsN e ∼ DBAbsN e ′

| DBAbsN e ∼ DBAbsV e ′

| DBAbsV e ∼ DBAbsN e ′

| DBAbsV e ∼ DBAbsV e ′

| DBFix e ∼ DBFix e ′

| DBtt ∼ DBtt
| DBff ∼ DBff
| DBNum n ∼ DBNum n

A program e2 refines the program e1 if it converges in context at least as often. This is a
preorder on programs.
definition

refines :: db ⇒ db ⇒ bool (‹- E -› [50 ,50 ] 50 )
where

e1 E e2 ≡ ∀C . ∃V1 . C<e1> ⇓ V1 −→ (∃V2 . C<e2> ⇓ V2 ∧ V1 ∼ V2 )

Contextually-equivalent programs refine each other.
definition

contextually-equivalent :: db ⇒ db ⇒ bool (‹- ≈ -›)
where

e1 ≈ e2 ≡ e1 E e2 ∧ e2 E e1

Our ultimate theorem states that if two programs have the same denotation then they are
contextually equivalent.
theorem computational-adequacy:

assumes 1 : closed e1
assumes 2 : closed e2
assumes D: evalDdb e1 ·env-empty-db = evalDdb e2 ·env-empty-db
shows e1 ≈ e2

This gives us a sound but incomplete method for demonstrating contextual equivalence. We
expect this result is useful for showing contextual equivalence for typed programs as well, but
leave it to future work to demonstrate this.
See Gunter (1992, §6.2) for further discussion of computational adequacy at higher types.
The reader may wonder why we did not use Nominal syntax to define our operational seman-
tics, following Urban and Narboux (2009). The reason is that Nominal2 does not support the
definition of continuous functions over Nominal syntax, which is required by the evaluators of
§3 and §4.1. As observed above, in the setting of traditional programming language semantics
one can get by with a much simpler notion of substitution than is needed for investigations
into λ-calculi. Clearly this does not hold of languages that reduce “under binders”.
The “fast and loose reasoning is morally correct” work of Danielsson et al. (2006) can be seen
as a kind of adequacy result.
Benton et al. (2009b) demonstrate a similar computational adequacy result in Coq. However
their system is only geared up for this kind of metatheory, and not reasoning about particular
programs; its term language is combinatory.
Benton et al. (2007, 2009a) have shown that it is difficult to scale this domain-theoretic

approach up to richer languages, such as those with dynamic allocation of mutable references,
especially if these references can contain (arbitrary) functional values.

5 Relating direct and continuation semantics

This is a fairly literal version of Reynolds (1974), adapted to untyped PCF. A more abstract
account has been given by Filinski (2007) in terms of a monadic meta language, which is
difficult to model in Isabelle (but see Huffman (2012a)).
We begin by giving PCF a continuation semantics following the modern account of Wadler
(1992). We use the symmetric function space ( ′o ValK , ′o) K → ( ′o ValK , ′o) K as our
language includes call-by-name.
type-synonym ( ′a, ′o) K = ( ′a → ′o) → ′o

domain ′o ValK
= ValKF (lazy appKF :: ( ′o ValK , ′o) K → ( ′o ValK , ′o) K )
| ValKTT | ValKFF
| ValKN (lazy nat)

type-synonym ′o ValKM = ( ′o ValK , ′o) K

We use the standard continuation monad to ease the semantic definition.
definition unitK :: ′o ValK → ′o ValKM where

unitK ≡ Λ a. Λ c. c·a

definition bindK :: ′o ValKM → ( ′o ValK → ′o ValKM ) → ′o ValKM where
bindK ≡ Λ m k. Λ c. m·(Λ a. k·a·c)

definition appKM :: ′o ValKM → ′o ValKM → ′o ValKM where
appKM ≡ Λ fK xK . bindK ·fK ·(Λ (ValKF ·f ). f ·xK )

The interpretations of the constants.
definition

condK :: ′o ValKM → ′o ValKM → ′o ValKM → ′o ValKM
where

condK ≡ Λ iK tK eK . bindK ·iK ·(Λ i. case i of
ValKF ·f ⇒ ⊥ | ValKTT ⇒ tK | ValKFF ⇒ eK | ValKN ·n ⇒ ⊥)

definition succK :: ′o ValKM → ′o ValKM where
succK ≡ Λ nK . bindK ·nK ·(Λ (ValKN ·n). unitK ·(ValKN ·(n + 1 )))

definition predK :: ′o ValKM → ′o ValKM where
predK ≡ Λ nK . bindK ·nK ·(Λ (ValKN ·n). case n of 0 ⇒ ⊥ | Suc n ⇒ unitK ·(ValKN ·n))

definition isZeroK :: ′o ValKM → ′o ValKM where
isZeroK ≡ Λ nK . bindK ·nK ·(Λ (ValKN ·n). unitK ·(if n = 0 then ValKTT else ValKFF))

A continuation semantics for PCF. If we had defined our direct semantics using a monad then
the correspondence would be more syntactically obvious.
type-synonym ′o EnvK = ′o ValKM Env

primrec
evalK :: expr ⇒ ′o EnvK → ′o ValKM

where
evalK (Var v) = (Λ %. %·v)

| evalK (App f x) = (Λ %. appKM ·(evalK f ·%)·(evalK x ·%))
| evalK (AbsN v e) = (Λ %. unitK ·(ValKF ·(Λ x. evalK e·(env-ext·v·x·%))))
| evalK (AbsV v e) = (Λ %. unitK ·(ValKF ·(Λ x c. x·(Λ x ′. evalK e·(env-ext·v·(unitK ·x ′)·%)·c))))
| evalK (Diverge) = (Λ %. ⊥)
| evalK (Fix v e) = (Λ %. µ x. evalK e·(env-ext·v·x·%))
| evalK (tt) = (Λ %. unitK ·ValKTT )
| evalK (ff ) = (Λ %. unitK ·ValKFF)
| evalK (Cond i t e) = (Λ %. condK ·(evalK i·%)·(evalK t·%)·(evalK e·%))
| evalK (Num n) = (Λ %. unitK ·(ValKN ·n))
| evalK (Succ e) = (Λ %. succK ·(evalK e·%))
| evalK (Pred e) = (Λ %. predK ·(evalK e·%))
| evalK (IsZero e) = (Λ %. isZeroK ·(evalK e·%))

To establish the chain completeness (admissibility) of our logical relation, we need to show
that unitK is an order monic, i.e., if unitK ·x v unitK ·y then x v y. This is an order-theoretic
version of injectivity.
In order to define a continuation that witnesses this, we need to be able to distinguish con-
verging and diverging computations. We therefore require our observation domain to contain
at least two elements:
locale at-least-two-elements =

fixes some-non-bottom-element :: ′o::domain
assumes some-non-bottom-element: some-non-bottom-element 6= ⊥

Following Reynolds (1974) and Filinski (2007, Remark 47) we use the following continuation:
lemma cont-below [simp, cont2cont]:

cont (λx:: ′a::pcpo. if x v d then ⊥ else c)

lemma (in at-least-two-elements) below-monic-unitK [intro, simp]:
below-monic-cfun (unitK :: ′o ValK → ′o ValKM )

proof(rule below-monicI )
fix v v ′ :: ′o ValK
assume vv ′: unitK ·v v unitK ·v ′

let ?k = Λ x. if x v v ′ then ⊥ else some-non-bottom-element
from vv ′ have unitK ·v·?k v unitK ·v ′·?k by (rule monofun-cfun-fun)
hence ?k·v v ?k·v ′ by (simp add: unitK-def )
with some-non-bottom-element show v v v ′ by (auto split: if-split-asm)

qed

5.1 Logical relation

We follow Reynolds (1974) by simultaneously defining a pair of relations over values and
functions. Both are bottom-reflecting, in contrast to the situation for computational adequacy
in §4.3. Filinski (2007) differs by assuming that values are always defined, and relates values
and monadic computations.
type-synonym ′o lfr = (ValD, ′o ValKM , ValD → ValD, ′o ValKM → ′o ValKM ) lf-pair-rep

type-synonym ′o lflf = (ValD, ′o ValKM , ValD → ValD, ′o ValKM → ′o ValKM ) lf-pair

context at-least-two-elements
begin

abbreviation lr-eta-rep-N where
lr-eta-rep-N ≡ { (e, e ′) .

(e = ⊥ ∧ e ′ = ⊥)
∨ (e = ValTT ∧ e ′ = unitK ·ValKTT )
∨ (e = ValFF ∧ e ′ = unitK ·ValKFF)
∨ (∃n. e = ValN ·n ∧ e ′ = unitK ·(ValKN ·n)) }

abbreviation lr-eta-rep-F where
lr-eta-rep-F ≡ λ(rm, rp). { (e, e ′) .

(e = ⊥ ∧ e ′ = ⊥)
∨ (∃ f f ′. e = ValF ·f ∧ e ′ = unitK ·(ValKF ·f ′) ∧ (f , f ′) ∈ unlr (snd rp)) }

definition lr-eta-rep where
lr-eta-rep ≡ λr . lr-eta-rep-N ∪ lr-eta-rep-F r

definition lr-theta-rep where
lr-theta-rep ≡ λ(rm, rp). { (f , f ′) .

(∀ (x, x ′) ∈ unlr (fst (undual rm)). (f ·x, f ′·x ′) ∈ unlr (fst rp)) }

definition lr-rep :: ′o lfr where
lr-rep ≡ λr . (lr-eta-rep r , lr-theta-rep r)

abbreviation lr :: ′o lflf where
lr ≡ λr . (mklr (fst (lr-rep r)), mklr (snd (lr-rep r)))end

It takes some effort to set up the minimal invariant relating the two pairs of domains. One
might hope this would be easier using deflations (which might compose) rather than “copy”
functions (which certainly don’t).
We elide these as they are tedious.
sublocale at-least-two-elements < F : DomSolP lr ValD-copy-rec ValK-copy-rec

apply standard
apply (rule mono-lr)

apply (rule fix-ValD-copy-rec-ID)
apply (rule fix-ValK-copy-rec-ID)

apply (simp-all add: cfun-map-def )[4 ]
apply (erule (2 ) min-inv-lr)
done

5.2 A retraction between the two definitions

We can use the relation to establish a strong connection between the direct and continuation
semantics. All results depend on the observation type being rich enough.
context at-least-two-elements
begin

abbreviation mrel (‹η: - 7→ -› [50 , 51 ] 50 ) where

η: x 7→ x ′ ≡ (x, x ′) ∈ unlr (fst F .delta)

abbreviation vrel (‹ϑ: - 7→ -› [50 , 51 ] 50 ) where
ϑ: y 7→ y ′ ≡ (y, y ′) ∈ unlr (snd F .delta)

Theorem 1 from Reynolds (1974).
lemma AbsV-aux:

assumes η: ValF ·f 7→ unitK ·(ValKF ·f ′)
shows η: ValF ·(strictify·f ) 7→ unitK ·(ValKF ·(Λ x c. x·(Λ x ′. f ′·(unitK ·x ′)·c)))

theorem Theorem1 :
assumes ∀ v. η: %·v 7→ % ′·v
shows η: evalD e·% 7→ evalK e·% ′

end

The retraction between the two value and monadic value spaces.
Note we need to work with an observation type that can represent the “explicit values”, i.e.
′o ValK.
locale value-retraction =

fixes VtoO :: ′o ValK → ′o
fixes OtoV :: ′o → ′o ValK
assumes OV : OtoV oo VtoO = ID

sublocale value-retraction < at-least-two-elements VtoO·(ValKN ·0 )
using OV by – (standard, simp add: injection-defined cfcomp1 cfun-eq-iff )

context value-retraction
begin

fun
DtoKM-i :: nat ⇒ ValD → ′o ValKM

and
KMtoD-i :: nat ⇒ ′o ValKM → ValD

where
DtoKM-i 0 = ⊥

| DtoKM-i (Suc n) = (Λ v. case v of
ValF ·f ⇒ unitK ·(ValKF ·(cfun-map·(KMtoD-i n)·(DtoKM-i n)·f ))

| ValTT ⇒ unitK ·ValKTT
| ValFF ⇒ unitK ·ValKFF
| ValN ·m ⇒ unitK ·(ValKN ·m))

| KMtoD-i 0 = ⊥
| KMtoD-i (Suc n) = (Λ v. case OtoV ·(v·VtoO) of

ValKF ·f ⇒ ValF ·(cfun-map·(DtoKM-i n)·(KMtoD-i n)·f )
| ValKTT ⇒ ValTT
| ValKFF ⇒ ValFF
| ValKN ·m ⇒ ValN ·m)

abbreviation DtoKM ≡ (
⊔

i. DtoKM-i i)
abbreviation KMtoD ≡ (

⊔
i. KMtoD-i i)

Lemma 1 from Reynolds (1974).

lemma Lemma1 :
η: x 7→ DtoKM ·x
η: x 7→ x ′ =⇒ x = KMtoD·x ′

Theorem 2 from Reynolds (1974).
theorem Theorem2 : evalD e·% = KMtoD·(evalK e·(DtoKM oo %))
using Lemma1 (2 )[OF Theorem1 ] Lemma1 (1 ) by (simp add: cfcomp1 )

end

Filinski (2007, Remark 48) observes that there will not be a retraction between direct and
continuation semantics for languages with richer notions of effects.
It should be routine to extend the above approach to the higher-order backtracking language
of Wand and Vaillancourt (2004).
I wonder if it is possible to construct continuation semantics from direct semantics as proposed
by Sethi and Tang (1980). Roughly we might hope to lift a retraction between two value
domains to a retraction at higher types by synthesising a suitable logical relation.

6 A small-step (reduction) operational semantics for PCF

A small-step semantics allows us to express more things, like the progress of well-typed pro-
grams.
FIXME adjust: This relation is non-deterministic, but only β-reduces terms where the argu-
ment is a value. Moreover if we start with a closed term then our values are also closed. So
while in general (i.e., for open terms) our substitution operation is wrong and this relation is
too big, we show that things work out if we start reducing from a closed term (i.e., a program).
FIXME following Tolmach https://www.cis.upenn.edu/~bcpierce/sf/current/Norm.html we
make this relation deterministic. Eases the normalization proof.
inductive

reduction :: db ⇒ db ⇒ bool (‹- →v -› [50 , 50 ] 50 )
where

betaN : DBApp (DBAbsN u) v →v u<v/0>
| betaV : val v =⇒ DBApp (DBAbsV u) v →v u<v/0>
| f →v f ′ =⇒ DBApp f x →v DBApp f ′ x
| [[f = DBAbsV u; x →v x ′]] =⇒ DBApp f x →v DBApp f x ′

| DBFix f →v f<DBFix f /0>
| DBCond DBtt t e →v t
| DBCond DBff t e →v e
| DBSucc (DBNum n) →v DBNum (Suc n)
| DBPred (DBNum (Suc n)) →v DBNum n
| DBIsZero (DBNum 0 ) →v DBtt
| 0 < n =⇒ DBIsZero (DBNum n) →v DBff

abbreviation — The transitive, reflexive closure of the reduction relation.
reduction-trc :: db ⇒ db ⇒ bool (‹- →v

∗ -› [100 , 100 ] 100 )
where

reduction-trc ≡ rtranclp reduction

declare reduction.intros[intro!]

inductive-cases reduction-inv:
DBVar v →v t ′

DBApp f x →v t ′

DBAbsN u →v t ′

DBAbsV u →v t ′

DBFix f →v t ′

DBCond i t e →v t ′

DBff →v t ′

DBtt →v t ′

DBNum n →v t ′

DBSucc n →v t ′

DBPred n →v t ′

DBIsZero n →v t ′

lemma reduction-val:
assumes val v
assumes v →v v ′

shows False
using assms by (auto elim: val.cases reduction-inv)

lemma reduction-deterministic:
assumes t →v t ′

assumes t →v t ′′

shows t ′′ = t ′

using assms by (induct arbitrary: t ′′) (blast dest: reduction-val elim: reduction-inv)+

6.0.1 Reduction is consistent with evaluation
lemma reduction-eval:

assumes t →v t ′

assumes t ′ ⇓ v
shows t ⇓ v

using assms by (induct arbitrary: v) (auto elim!: evalOP-inv val.cases intro: eval-val)

lemma reduction-trc-eval:
assumes t →v

∗ t ′

assumes t ′ ⇓ v
shows t ⇓ v

using assms by induct (auto simp: reduction-eval)

theorem reduction-trc-val-eval:
assumes t →v

∗ v
assumes val v
shows t ⇓ v

using assms by (induct rule: converse-rtranclp-induct) (auto intro: eval-val reduction-trc-eval)

We show the converse (of sorts) using the frame stack machinery of the next section.

7 Concluding remarks

We have seen that Pitts’s techniques for showing the existence of relations over domains is
straightforward to mechanise and use in HOLCF.
One source of irritation in doing so is that Pitts’s technique is formulated in terms of minimal
invariants, which presently must be written out by hand. (Earlier versions of HOLCF’s
domain package provided these copy functions, though we would still need to provide our
own in such cases as §5.) HOLCF ’11 provides us with take functions (approximations,
deflations) on domains that compose, and so one might hope to adapt Pitts’s technique to
use these instead. This has been investigated by Benton et al. (2009a, §6), but it is unclear
that the deflations involved are those generated by HOLCF ’11.

